```
innerproduct[v1_, v2_] := 1 / 2 Integrate[v1 * Conjugate[v2], {t, -1, 1}];
norm[v_] := Sqrt[innerproduct[v, v]];
```

General::spell1 : Possible spelling error: new symbol name "norm" is similar to existing symbol "Norm". More…

```
u0 = 1;
u1 = t;
u2 = t^2;
u3 = t^3;
```

```
Plot[{u0, u1, u2, u3}, {t, -1, 1}, PlotStyle → {Black, Red, Green, Blue}, ImageSize → 600];
```




```
tmp = u0
ϕ0 = tmp / norm[tmp] // Simplify
```

1

1

```
tmp = u1
tmp -= innerproduct[tmp, ϕ0] ϕ0
ϕ1 = tmp / norm[tmp] // Simplify
```

t

t

$\sqrt{3}\, t$

```
tmp = u2
tmp -= innerproduct[tmp, ϕ0] ϕ0
tmp -= innerproduct[tmp, ϕ1] ϕ1
ϕ2 = tmp / norm[tmp] // Simplify
```

t^2

$t^2 - \frac{1}{3}$

$t^2 - \frac{1}{3}$

$\frac{1}{2} \sqrt{5} (3 t^2 - 1)$

```
tmp = u3
tmp -= innerproduct[tmp, ϕ0] ϕ0
tmp -= innerproduct[tmp, ϕ1] ϕ1
tmp -= innerproduct[tmp, ϕ2] ϕ2
ϕ3 = tmp / norm[tmp] // Simplify
```

t^3

t^3

$t^3 - \frac{3 t}{5}$

$t^3 - \frac{3 t}{5}$

$\frac{1}{2} \sqrt{7} \, t (5 t^2 - 3)$

```
Plot[{ϕ0, ϕ1, ϕ2, ϕ3}, {t, -1, 1}, PlotStyle → {Black, Red, Green, Blue}, ImageSize → 600];
```




```
{
 {innerproduct[ϕ0, ϕ0],
  innerproduct[ϕ0, ϕ1],
  innerproduct[ϕ0, ϕ2],
  innerproduct[ϕ0, ϕ3]},
 {innerproduct[ϕ1, ϕ0],
  innerproduct[ϕ1, ϕ1],
  innerproduct[ϕ1, ϕ2],
  innerproduct[ϕ1, ϕ3]},
 {innerproduct[ϕ2, ϕ0],
  innerproduct[ϕ2, ϕ1],
  innerproduct[ϕ2, ϕ2],
  innerproduct[ϕ2, ϕ3]},
 {innerproduct[ϕ3, ϕ0],
  innerproduct[ϕ3, ϕ1],
  innerproduct[ϕ3, ϕ2],
  innerproduct[ϕ3, ϕ3]}
}
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
x = Sin[π t]

xhat =
 innerproduct[x, ϕ0] ϕ0 +
  innerproduct[x, ϕ1] ϕ1 +
  innerproduct[x, ϕ2] ϕ2 +
  innerproduct[x, ϕ3] ϕ3


xtaylor =
 (D[x, {t, 0}] /. t → 0) / 0! * 1 +
  (D[x, {t, 1}] /. t → 0) / 1! * (t - 0) +
  (D[x, {t, 2}] /. t → 0) / 2! * (t - 0) ^2 +
  (D[x, {t, 3}] /. t → 0) / 3! * (t - 0) ^3
```

$\sin(\pi t)$

$$\frac{7(-15+\pi^2)(5t^2-3)t}{2\pi^3}+\frac{3t}{\pi}$$

$$\pi t-\frac{\pi^3 t^3}{6}$$

```
Plot[{x, xhat, xtaylor}, {t, -1, 1}, PlotStyle → {Black, Red, Blue}, ImageSize → 600];
```



```
Plot[{x - xhat, x - xtaylor}, {t, -1, 1}, PlotStyle → {Red, Blue}, ImageSize → 600];
```




```
norm[x - xhat]^2 // N
```

0.00439012

The Taylor series is a much better approximation near the point of expansion t = 0, but is relatively bad away from t = 0. The minimum mean square error (MMSE) approximation spreads the error out over the interval, and isn't as good as the Taylor series is near 0, but isn't as bad near the edges.